



# **A Side-by-Side Comparison of eCQM Logic Expression Using Clinical Quality Language Versus the Quality Data Model for Eligible Hospitals and Critical Access Hospitals (CAHs)**

**Monday, December 18, 2017  
1:00-2:00 PM ET**

---

**Shanna Hartman**  
**Centers for Medicare & Medicaid Services**  
**and**  
**Bryn Rhodes, ESAC, Inc.**

# Agenda

- Welcome and Background
- What is Clinical Quality Language (CQL)?
- Measure Tour – CMS 104
- Measure Tour – CMS 55
- Available Tools and Resources


# Background

- CQL is a Health Level Seven International (HL7) standard and aims to unify the expression of logic for electronic clinical quality measures (eCQMs) and Clinical Decision Support (CDS).
- CQL provides the ability to better express logic defining measure populations to improve the accuracy and clarity of eCQMs.
- Benefits of CQL:
  - Improved expressivity
  - More precise/unambiguous
  - Can share logic between measures
  - Can share logic with decision support
  - Can be used with multiple information data models (e.g., Quality Data Model [QDM], Fast Healthcare Interoperability Resources [FHIR])
  - Simplifies calculation engine implementation

# CQL Transition

- eCQMs will be transitioned to use the CQL standard for logic expression
- The transition will begin with the CY 2019 reporting period for Eligible Hospitals and Critical Access Hospitals (CAHs), and CY 2019 performance period for Eligible Professionals (EPs) and Eligible Clinicians for the following programs:
  - Hospital Inpatient Quality Reporting Program
  - Medicare Electronic Health Record Incentive Program for Eligible Hospitals and CAHs
  - Medicaid EHR Incentive Program for EPs, Eligible Hospitals, and CAHs
  - Quality Payment Program: The Merit-based Incentive Payment System (MIPS) and Alternative Payment Models
- To support the transition, CMS will publish CQL-based eCQMs in Spring 2018

# Evolving eCQM Standards



**Definitions:**


**HQMF** – Health Quality Measure Format

**CQL** – Clinical Quality Language

**QDM** – Quality Data Model



# Measure Development – Expected Timelines



a – Measure Structure – HQMF  
b – CQL-based HQMF  
c – Measure Report – QRDA Category I  
(individual report), QRDA Category III  
(aggregate report)

---

# WHAT IS CQL?

# Quality Measurement

- What is a quality measure?
  - Quantitative tool to assess performance related to a specific clinical process or outcome
- What is an electronic Clinical Quality Measure (eCQM)?
  - Electronic representation of a quality measure with the goal of enabling the measure to be evaluated as automatically as possible

# CMS 104 Draft – Description

Ischemic stroke patients prescribed or continuing to take antithrombotic therapy at hospital discharge.

Questions “about” the description:

- Who said it?
- When did they say it?
- What evidence supports it?
- How should I use it?

Questions about the content of the description:


- What kinds of “things” does it talk about?
- What do those “things” look like?
- What are the relationships between them?
- What are the criteria that apply to them?

Metadata


Data Model

Logic

# eCQM Representation




# QDM Data Types



# Encounter, Performed

- Data elements that meet criteria using this datatype should document that the encounter indicated by the QDM category and its corresponding value set is in progress or has been completed.
- Encounter, Performed has the following attributes:
  - Id
  - Code
  - Relevant Period
  - Admission Source
  - Diagnoses
  - Discharge Disposition
  - Length of Stay
  - Negation Rationale
  - Principal Diagnosis
  - Author Datetime
  - Facility Locations

# Components of Sharing Logic



# What is CQL?

A standard language for expressing clinical knowledge that is

- Readable
- Shareable
- Computable

---

# MEASURE TOURS

# Measure Tours

- A side-by-side review of the
  - Current eCQM specification using QDM 4.3
  - New eCQM specification using QDM 5.3 and CQL
- Tour of two measures:
  - CMS 104 – Discharged on Antithrombotics
  - CMS 55 – Median ED Arrival to Departure
- NOTE: These draft CQL eCQM specifications are for informational review only

---

# MEASURE TOUR – CMS 104

# Measure Package

## QDM-Based

- .html
- .xml
- \_\_SimpleXML.xml

## CQL-Based

- \_\_HumanReadable.html
- \_\_eCQM.xml
- \_\_CQL.cql
- \_\_ELM.xml
- \_\_ELM.json

NOTE: File-naming conventions within measure packages are still being finalized, and will be posted to the eCQI Resource Center once available. This listing is only to illustrate the types of files that will be present in CQL-based measure packages, and how they compare with the types of files that were present in QDM-based measure packages.

# Measure Metadata

eMeasure Title	Discharged on Antithrombotic Therapy		
eMeasure Identifier (Measure Authoring Tool)	104	eMeasure Version number	6.1.000
NQF Number	Not Applicable	GUID	42bf391f-38a3-4c0f-9ece-dcd47e9609d9
Measurement Period	January 1, 20XX through December 31, 20XX		
Measure Steward	The Joint Commission		
Measure Developer	The Joint Commission		
Endorsed By	None		
Description	Ischemic stroke patients prescribed or continuing to take antithrombotic therapy at hospital discharge		
Copyright	Measure specifications are in the Public Domain  LOINC(R) is a registered trademark of the Regenstrief Institute.  This material contains SNOMED Clinical Terms (R) (SNOMED CT(C)) copyright 2004-2016 International Health Terminology Standards Development Organization. All rights reserved.		
Disclaimer	These performance measures are not clinical guidelines and do not establish a standard of medical care, and have		

# Measure Contents

## QDM-Based

- Population Criteria
- Data Criteria (QDM Variables)
- Data Criteria (QDM Data Elements)
- Supplemental Data Elements
- Risk Adjustments Variables

## CQL-Based

- Population Criteria
- Definitions
- Functions
- Terminology
- Data Criteria (QDM Data Elements)
- Supplemental Data Elements
- Risk Adjustments Variables

# Population Criteria

## Population Criteria

- **Initial Population** =
  - AND: Age >= 18 year(s) at: Occurrence A of \$EncounterInpatientNonElective
  - AND:
    - OR: Intersection of:
      - Occurrence A of \$EncounterInpatientNonElective
      - "Encounter, Performed: Non-Elective Inpatient Encounter (principal diagnosis: Ischemic Stroke)"
    - OR: Intersection of:
      - Occurrence A of \$EncounterInpatientNonElective
      - "Encounter, Performed: Non-Elective Inpatient Encounter (principal diagnosis: Hemorrhagic Stroke)"
- **Denominator** =
  - AND: Initial Population
  - AND: Intersection of:
    - Occurrence A of \$EncounterInpatientNonElective
    - "Encounter, Performed: Non-Elective Inpatient Encounter (principal diagnosis: Ischemic Stroke)"
- **Denominator Exclusions** =
  - OR: Intersection of:
    - Occurrence A of \$EncounterInpatientNonElective
    - "Encounter, Performed: Non-Elective Inpatient Encounter" satisfies any:
      - (discharge status: Discharge To Acute Care Facility)
      - (discharge status: Left Against Medical Advice)
      - (discharge status: Patient Expired)
      - (discharge status: Discharged to Home for Hospice Care)
      - (discharge status: Discharged to Health Care Facility for Hospice Care)
  - OR: \$InterventionComfortMeasures starts during ("Encounter, Performed: Emergency Department Visit" <= 1 hour(s) ends before or concurrent with start of Occurrence A of \$EncounterInpatientNonElective )
  - OR: \$InterventionComfortMeasures starts during Occurrence A of \$EncounterInpatientNonElective
- **Numerator** =
  - AND: "Medication, Discharge: Antithrombotic Therapy" starts during Occurrence A of \$EncounterInpatientNonElective
- **Numerator Exclusions** =
  - None
- **Denominator Exceptions** =
  - OR: Union of:
    - "Medication, Discharge not done: Medical Reason" for "Antithrombotic Therapy"
    - "Medication, Discharge not done: Patient Refusal" for "Antithrombotic Therapy"
    - starts during Occurrence A of \$EncounterInpatientNonElective
- **Stratification** =
  - None

## Population Criteria

### Initial Population

TJC."Encounter with Principal Diagnosis and Age"

### Denominator

TJC."Encounter with Principal Diagnosis of Ischemic Stroke"

### Denominator Exclusions

(  
TJC."Encounter with Principal Diagnosis of Ischemic Stroke" Encounter  
where Encounter.dischargeDisposition in "Discharge To Acute Care Facility"  
or Encounter.dischargeDisposition in "Left Against Medical Advice"  
or Encounter.dischargeDisposition in "Patient Expired"  
or Encounter.dischargeDisposition in "Discharged to Home for Hospice Care"  
or Encounter.dischargeDisposition in "Discharged to Health Care Facility for Hospice Care"  
)  
union "Comfort Measures during Hospitalization"

### Numerator

TJC."Encounter with Principal Diagnosis of Ischemic Stroke" Encounter  
with "Antithrombotic Therapy at Discharge" Antithrombotic  
such that Antithrombotic.authorDatetime during Encounter.relevantPeriod

### Numerator Exclusions

None

### Denominator Exceptions

TJC."Encounter with Principal Diagnosis of Ischemic Stroke" Encounter  
with "Antithrombotic Not Given at Discharge" Antithrombotic  
such that Antithrombotic.authorDatetime during Encounter.relevantPeriod

### Stratification

None

# CQL Library

```
1  library Test104 version '2.0.002'
2
3  using QDM version '5.3'
4
5  include MATGlobalCommonFunctions version '2.0.000' called global
6  include TJC_Overall version '7.0.000' called TJC
7
8  valueset "ONC Administrative Sex": 'urn:oid:2.16.840.1.113762.1.4.1'
9  valueset "Race": 'urn:oid:2.16.840.1.114222.4.11.836'
10 valueset "Ethnicity": 'urn:oid:2.16.840.1.114222.4.11.837'
11 valueset "Payer": 'urn:oid:2.16.840.1.114222.4.11.3591'
12 valueset "Comfort Measures": 'urn:oid:1.3.6.1.4.1.33895.1.3.0.45'
13 valueset "Antithrombotic Therapy": 'urn:oid:2.16.840.1.113883.3.117.1.7.1.201'
14 valueset "Patient Expired": 'urn:oid:2.16.840.1.113883.3.117.1.7.1.309'
15 valueset "Left Against Medical Advice": 'urn:oid:2.16.840.1.113883.3.117.1.7.1.308'
16 valueset "Discharged to Health Care Facility for Hospice Care": 'urn:oid:2.16.840.1.113883.3.117.1.7.1.207'
17 valueset "Discharge To Acute Care Facility": 'urn:oid:2.16.840.1.113883.3.117.1.7.1.87'
18 valueset "Discharged to Home for Hospice Care": 'urn:oid:2.16.840.1.113883.3.117.1.7.1.209'
19 valueset "Medical Reason": 'urn:oid:2.16.840.1.113883.3.117.1.7.1.473'
20 valueset "Patient Refusal": 'urn:oid:2.16.840.1.113883.3.117.1.7.1.93'
21
22 context Patient
23
```

# Initial Population

## QDM-Based

- **Initial Population =**
  - AND: Age >= 18 year(s) at: Occurrence A of \$EncounterInpatientNonElective
  - AND:
    - OR: Intersection of:
      - Occurrence A of \$EncounterInpatientNonElective
      - "Encounter, Performed: Non-Elective Inpatient Encounter (principal diagnosis: Ischemic Stroke)"
    - OR: Intersection of:
      - Occurrence A of \$EncounterInpatientNonElective
      - "Encounter, Performed: Non-Elective Inpatient Encounter (principal diagnosis: Hemorrhagic Stroke)"

## CQL-Based

### Initial Population

TJC."Encounter with Principal Diagnosis and Age"

### TJC.Encounter with Principal Diagnosis and Age

"Encounter with Principal Diagnosis of Hemorrhagic or Ischemic Stroke" NonElectiveEncounter  
with ["Patient Characteristic Birthdate"] P  
such that global."CalendarAgeInYearsAt"(P.birthDatetime, start of NonElectiveEncounter.relevantPeriod)>= 18

# CQL Expressions

- Logic
  - A and B
  - A and not (B or C)
- Comparison
  - A >= B
  - A <> B
- Arithmetic
  - A + B
  - A + B \* C

# Non Elective Inpatient Encounter

## QDM-Based

- `$EncounterInpatientNonElective =`
  - "Encounter, Performed: Non-Elective Inpatient Encounter" satisfies all:
    - (length of stay <= 120 day(s))
    - ends during "Measurement Period"

## CQL-Based

### ▲ `TJC.Non Elective Inpatient Encounter`

```
["Encounter, Performed": "Non-Elective Inpatient Encounter"] NonElectiveEncounter  
where NonElectiveEncounter.lengthOfStay <= 120 days  
and NonElectiveEncounter.relevantPeriod ends during "Measurement Period"
```

In CQL, specific occurrences are no longer required. Instead, we define an expression that identifies the specific occurrence and use that definition in subsequent criteria.

# Retrieve (square brackets)



- The “type” portion must be the name of some type defined by the model
  - QDM version 5.3 in this case
  - Encounter is the QDM “category”, Encounter, Performed is the datatype
- The “terminology” portion must be a valueset, code, or codesystem
- Result is the set of data elements of the specified *type* that have a *code* that matches the terminology

# Non Elective Inpatient Encounter

## ▲ TJC.Non Elective Inpatient Encounter

[ "Encounter, Performed": "Non-Elective Inpatient Encounter" ] NonElectiveEncounter  
where NonElectiveEncounter.lengthOfStay <= 120 days  
and NonElectiveEncounter.relevantPeriod ends during "Measurement Period"

- The result of the retrieve is then a *set* of encounters, as opposed to a “yes/no”
- Sets are combined with “intersect” and “union”
  - vs criteria, which are combined with “and” and “or”
- This is a *query*, which is introduced with the “NonElectiveEncounter” *alias*
- The where clause can then use this alias to talk about each encounter in the result

# Filtering with Where

where NonElectiveEncounter.lengthOfStay <= 120 days  
and NonElectiveEncounter.relevantPeriod ends during "Measurement Period"

- “NonElectiveEncounter” refers to the encounters in the “source” of the query
  - In this case, encounters with a code in the appropriate value set
- “Encounter, Performed” structure (i.e. the attributes) is defined by QDM 5.3
  - In this case, the datatype defines attributes such as “relevantPeriod”
- “relevantPeriod” and “Measurement Period” are both DateTime *intervals*
  - Meaning they have a start and end point that is a DateTime value
- CQL supports interval comparisons like this directly
  - e.g. “A during B”, “A overlaps B”, or “A includes B”
- CQL also supports timing phrases
  - e.g. “A starts before start B” or “A starts 1 day or less after end B”

# Timing Relationships

Comparing two date/time values

```
Encounter.authorDatetime < assessment.authorDatetime
```

Comparing a date/time value with an interval (period)

```
assessment.authorDatetime during Encounter.relevantPeriod
```

Comparing an interval with a date/time value

```
Encounter.relevantPeriod includes assessment.authorDatetime
```

Comparing two intervals

```
Encounter.relevantPeriod during "Measurement Period"
```

# Timing and Intervals in CQL

- Full set from QDM
  - starts before start, starts same day as
- Timing phrases
  - starts 3 days before start
  - starts 3 days or less before start
  - starts within 3 days of start
- Interval operators
  - meets, overlaps, during
- Boundary access
  - start of MeasurementPeriod
- Membership
  - X in interval[4, 6]

# Denominator

## QDM-Based

- **Denominator =**
  - AND: Initial Population
  - AND: Intersection of:
    - Occurrence A of \$EncounterInpatientNonElective
    - "Encounter, Performed: Non-Elective Inpatient Encounter (principal diagnosis: Ischemic Stroke)"

## CQL-Based

### ▲ Denominator

TJC."Encounter with Principal Diagnosis of Ischemic Stroke"

### ▲ TJC.Encounter with Principal Diagnosis of Ischemic Stroke

"Encounter with Principal Diagnosis and Age" NonElectiveEncounter  
where NonElectiveEncounter.principalDiagnosis in "Ischemic Stroke"

# Measure Libraries

- CQL allows definitions to be shared among measures using Libraries
- “TJC” in this case is a library defining several expressions that are shared among multiple measures
- Measure packages will include artifacts for any libraries they reference

# Denominator Exclusions

## QDM-Based

- **Denominator Exclusions =**
  - OR: Intersection of:
    - Occurrence A of \$EncounterInpatientNonElective
    - "Encounter, Performed: Non-Elective Inpatient Encounter" satisfies any:
      - (discharge status: Discharge To Acute Care Facility)
      - (discharge status: Left Against Medical Advice)
      - (discharge status: Patient Expired)
      - (discharge status: Discharged to Home for Hospice Care)
      - (discharge status: Discharged to Health Care Facility for Hospice Care)
  - OR: \$InterventionComfortMeasures starts during ("Encounter, Performed: Emergency Department Visit" <= 1 hour(s) ends before or concurrent with start of Occurrence A of \$EncounterInpatientNonElective )
  - OR: \$InterventionComfortMeasures starts during Occurrence A of \$EncounterInpatientNonElective

## CQL-Based

### ▲ Denominator Exclusions

```
(
  TJC."Encounter with Principal Diagnosis of Ischemic Stroke" Encounter
  where Encounter.dischargeDisposition in "Discharge To Acute Care Facility"
  or Encounter.dischargeDisposition in "Left Against Medical Advice"
  or Encounter.dischargeDisposition in "Patient Expired"
  or Encounter.dischargeDisposition in "Discharged to Home for Hospice Care"
  or Encounter.dischargeDisposition in "Discharged to Health Care Facility for Hospice Care"
)
union "Comfort Measures during Hospitalization"
```

# Comfort Measures During Hospitalization

## QDM-Based

- OR: `$InterventionComfortMeasures` starts during ("Encounter, Performed: Emergency Department Visit" `<= 1 hour(s)` ends before or concurrent with start of Occurrence A of `$EncounterInpatientNonElective` )
- OR: `$InterventionComfortMeasures` starts during Occurrence A of `$EncounterInpatientNonElective`

## CQL-Based

### ▲ Comfort Measures during Hospitalization

TJC."Encounter with Principal Diagnosis of Ischemic Stroke" `NonElectiveEncounter`  
with "Intervention Comfort Measures" `ComfortMeasure`  
such that `Coalesce(start of ComfortMeasure.relevantPeriod, ComfortMeasure.authorDatetime)`  
during `global."Hospitalization"(NonElectiveEncounter)`

# Numerator

## QDM-Based

- **Numerator** =
  - AND: "Medication, Discharge: Antithrombotic Therapy" starts during Occurrence A of \$EncounterInpatientNonElective

## CQL-Based

### ▲ Numerator

TJC."Encounter with Principal Diagnosis of Ischemic Stroke" Encounter  
with "Antithrombotic Therapy at Discharge" Antithrombotic  
such that Antithrombotic.authorDatetime during Encounter.relevantPeriod

### ▲ TJC.Encounter with Principal Diagnosis of Ischemic Stroke

"Encounter with Principal Diagnosis and Age" NonElectiveEncounter  
where NonElectiveEncounter.principalDiagnosis in "Ischemic Stroke"

### ▲ Antithrombotic Therapy at Discharge

["Medication, Discharge": "Antithrombotic Therapy"] Antithrombotic

# Denominator Exceptions

## QDM-Based

- **Denominator Exceptions =**
  - OR: Union of:
    - "Medication, Discharge not done: Medical Reason" for "Antithrombotic Therapy"
    - "Medication, Discharge not done: Patient Refusal" for "Antithrombotic Therapy"
    - starts during Occurrence A of \$EncounterInpatientNonElective

## CQL-Based

### ▲ Denominator Exceptions

TJC."Encounter with Principal Diagnosis of Ischemic Stroke" Encounter  
with "Antithrombotic Not Given at Discharge" Antithrombotic  
such that Antithrombotic.authorDatetime during Encounter.relevantPeriod

### ▲ Antithrombotic Not Given at Discharge

["Medication, Not Discharged": "Antithrombotic Therapy"] Antithrombotic  
where Antithrombotic.negationRationale in "Medical Reason"  
or Antithrombotic.negationRationale in "Patient Refusal"

# Data Criteria

## QDM-Based

- "Encounter, Performed: Emergency Department Visit" using "Emergency Department Visit SNOMEDCT Value Set (2.16.840.1.113883.3.117.1.7.1.292)"
- "Encounter, Performed: Non-Elective Inpatient Encounter" using "Non-Elective Inpatient Encounter SNOMEDCT Value Set (2.16.840.1.113883.3.117.1.7.1.424)"
- "Intervention, Order: Comfort Measures" using "Comfort Measures SNOMEDCT Value Set (1.3.6.1.4.1.33895.1.3.0.45)"
- "Intervention, Performed: Comfort Measures" using "Comfort Measures SNOMEDCT Value Set (1.3.6.1.4.1.33895.1.3.0.45)"
- "Medication, Discharge: Antithrombotic Therapy" using "Antithrombotic Therapy RXNORM Value Set (2.16.840.1.113883.3.117.1.7.1.201)"
- "Medication, Discharge not done: Medical Reason" using "Medical Reason SNOMEDCT Value Set (2.16.840.1.113883.3.117.1.7.1.473)"
- "Medication, Discharge not done: Patient Refusal" using "Patient Refusal SNOMEDCT Value Set (2.16.840.1.113883.3.117.1.7.1.93)"
- Attribute: "Discharge status: Patient Expired" using "Patient Expired SNOMEDCT Value Set (2.16.840.1.113883.3.117.1.7.1.309)"
- Attribute: "Discharge status: Left Against Medical Advice" using "Left Against Medical Advice SNOMEDCT Value Set (2.16.840.1.113883.3.117.1.7.1.308)"
- Attribute: "Discharge status: Discharged to Health Care Facility for Hospice Care" using "Discharged to Health Care Facility for Hospice Care SNOMEDCT Value Set (2.16.840.1.113883.3.117.1.7.1.207)"
- Attribute: "Principal diagnosis: Ischemic Stroke" using "Ischemic Stroke Grouping Value Set (2.16.840.1.113883.3.117.1.7.1.247)"
- Attribute: "Discharge status: Discharge To Acute Care Facility" using "Discharge To Acute Care Facility SNOMEDCT Value Set (2.16.840.1.113883.3.117.1.7.1.87)"
- Attribute: "Discharge status: Discharged to Home for Hospice Care" using "Discharged to Home for Hospice Care SNOMEDCT Value Set (2.16.840.1.113883.3.117.1.7.1.209)"
- Attribute: "Principal diagnosis: Hemorrhagic Stroke" using "Hemorrhagic Stroke Grouping Value Set (2.16.840.1.113883.3.117.1.7.1.212)"

## CQL-Based

- "Encounter, Performed: Emergency Department Visit" using "Emergency Department Visit (2.16.840.1.113883.3.117.1.7.1.292)"
- "Encounter, Performed: Non-Elective Inpatient Encounter" using "Non-Elective Inpatient Encounter (2.16.840.1.113883.3.117.1.7.1.424)"
- "Intervention, Order: Comfort Measures" using "Comfort Measures (1.3.6.1.4.1.33895.1.3.0.45)"
- "Intervention, Performed: Comfort Measures" using "Comfort Measures (1.3.6.1.4.1.33895.1.3.0.45)"
- "Medication, Discharge: Antithrombotic Therapy" using "Antithrombotic Therapy (2.16.840.1.113883.3.117.1.7.1.201)"
- "Medication, Not Discharged: Antithrombotic Therapy" using "Antithrombotic Therapy (2.16.840.1.113883.3.117.1.7.1.201)"
- "Patient Characteristic Ethnicity: Ethnicity" using "Ethnicity (2.16.840.1.114222.4.11.837)"
- "Patient Characteristic Payer: Payer" using "Payer (2.16.840.1.114222.4.11.3591)"
- "Patient Characteristic Race: Race" using "Race (2.16.840.1.114222.4.11.836)"
- "Patient Characteristic Sex: ONC Administrative Sex" using "ONC Administrative Sex (2.16.840.1.113762.1.4.1)"

# Terminology

## CQL-Based (only)

### Terminology

- valueset "Antithrombotic Therapy" using "2.16.840.1.113883.3.117.1.7.1.201"
  - valueset "Comfort Measures" using "1.3.6.1.4.1.33895.1.3.0.45"
  - valueset "Discharge To Acute Care Facility" using "2.16.840.1.113883.3.117.1.7.1.87"
  - valueset "Discharged to Health Care Facility for Hospice Care" using "2.16.840.1.113883.3.117.1.7.1.207"
  - valueset "Discharged to Home for Hospice Care" using "2.16.840.1.113883.3.117.1.7.1.209"
  - valueset "Emergency Department Visit" using "2.16.840.1.113883.3.117.1.7.1.292"
  - valueset "Ethnicity" using "2.16.840.1.114222.4.11.837"
  - valueset "Hemorrhagic Stroke" using "2.16.840.1.113883.3.117.1.7.1.212"
  - valueset "Ischemic Stroke" using "2.16.840.1.113883.3.117.1.7.1.247"
  - valueset "Left Against Medical Advice" using "2.16.840.1.113883.3.117.1.7.1.308"
  - valueset "Medical Reason" using "2.16.840.1.113883.3.117.1.7.1.473"
  - valueset "Non-Elective Inpatient Encounter" using "2.16.840.1.113883.3.117.1.7.1.424"
  - valueset "ONC Administrative Sex" using "2.16.840.1.113762.1.4.1"
  - valueset "Patient Expired" using "2.16.840.1.113883.3.117.1.7.1.309"
  - valueset "Patient Refusal" using "2.16.840.1.113883.3.117.1.7.1.93"
  - valueset "Payer" using "2.16.840.1.114222.4.11.3591"
  - valueset "Race" using "2.16.840.1.114222.4.11.836"
- Includes all terminologies referenced by the measure
  - This may include direct-reference codes now, rather than only value sets

# Supplemental Data

## QDM-Based

- "Patient Characteristic Ethnicity: Ethnicity" using "Ethnicity CDCREC Value Set (2.16.840.1.114222.4.11.837)"
- "Patient Characteristic Payer: Payer" using "Payer SOP Value Set (2.16.840.1.114222.4.11.3591)"
- "Patient Characteristic Race: Race" using "Race CDCREC Value Set (2.16.840.1.114222.4.11.836)"
- "Patient Characteristic Sex: ONC Administrative Sex" using "ONC Administrative Sex AdministrativeGender Value Set (2.16.840.1.113762.1.4.1)"

## CQL-Based

### ▲ SDE Ethnicity

["Patient Characteristic Ethnicity": "Ethnicity"]

### ▲ SDE Payer

["Patient Characteristic Payer": "Payer"]

### ▲ SDE Race

["Patient Characteristic Race": "Race"]

### ▲ SDE Sex

["Patient Characteristic Sex": "ONC Administrative Sex"]

---

# MEASURE TOUR – CMS 55

# Continuous Variable Measure

## Continuous Variable (CMS55)

Median time from emergency department admission to time of discharge from the emergency room for patients admitted to the facility from the emergency department

- Calculates an “observation” for each member of the population
- As opposed to a proportion measure that calculates a percentage using the sizes of the numerator and denominator

# Initial Population

## QDM-Based

- **Initial Population =**
  - AND: "Occurrence A of Encounter, Performed: Emergency Department Visit" <= 1 hour(s) ends before or concurrent with start of Occurrence A of \$EncounterInpatient

## CQL-Based

### ▲ Initial Population

"Inpatient Encounter" Encounter  
with ["Encounter, Performed": "Emergency department patient visit (procedure)"]ED  
such that ED.relevantPeriod ends 1 hour or less before or on start of Encounter.relevantPeriod

# Measure Observation

## QDM-Based

- **Measure Observation =**
  - Median:
    - Datetimediff:
      - "Occurrence A of Encounter, Performed: Emergency Department Visit (facility location departure datetime)"
      - "Occurrence A of Encounter, Performed: Emergency Department Visit (facility location arrival datetime)"

## CQL-Based

### ▲ Measure Observation

duration in minutes of "Arrival And Departure Time"("Related ED Visit"(Encounter))

### ▲ Related ED Visit(Encounter "Encounter, Performed")

Last(["Encounter, Performed": "Emergency department patient visit (procedure)"]ED  
where ED.relevantPeriod ends 1 hour or less before or on start of Encounter.relevantPeriod  
sort by start of relevantPeriod)

# Stratifiers

## QDM-Based

- **Stratification 1** =
  - AND NOT: Intersection of:
    - Occurrence A of \$EncounterInpatient
    - "Encounter, Performed: Encounter Inpatient (principal diagnosis: Psychiatric/Mental Health Patient)"
- **Stratification 2** =
  - AND: Intersection of:
    - Occurrence A of \$EncounterInpatient
    - "Encounter, Performed: Encounter Inpatient (principal diagnosis: Psychiatric/Mental Health Patient)"

## CQL-Based

### ▲ Stratum1

"Inpatient Encounter" Encounter  
where not ( Encounter.principalDiagnosis in "Psychiatric/Mental Health Patient" )

### ▲ Stratum2

"Inpatient Encounter" Encounter  
where Encounter.principalDiagnosis in "Psychiatric/Mental Health Patient"

# Available Tools and Resources

- CQL Specification - CQL Release 1, Standard for Trial Use (STU) 2
  - [http://www.hl7.org/implement/standards/product\\_brief.cfm?product\\_id=400](http://www.hl7.org/implement/standards/product_brief.cfm?product_id=400)
- CQL-Based HQMF IG – Release 1, STU 2.1
  - [http://www.hl7.org/implement/standards/product\\_brief.cfm?product\\_id=405](http://www.hl7.org/implement/standards/product_brief.cfm?product_id=405)
- eCQI Resource Center
  - CQL Space, including the QDM v5.3 and v5.3 Annotated
    - <https://ecqi.healthit.gov/cql>
  - Check the eCQI Resource Center Events page and CQL Educational Resources page for more information
    - <https://ecqi.healthit.gov/ecqi/ecqi-events>
    - <https://ecqi.healthit.gov/cql/cql-educational-resources>



# Available Tools and Resources (Cont'd)

- CQL Formatting and Usage Wiki
  - <https://github.com/esacinc/CQL-Formatting-and-Usage-Wiki/wiki>
- CQL GitHub Tools Repository
  - [https://github.com/cqframework/clinical\\_quality\\_language](https://github.com/cqframework/clinical_quality_language)
- Measure Authoring Tool
  - <https://www.emeasuretool.cms.gov/>
- Bonnie Testing Tool
  - <https://bonnie.healthit.gov/>
- To submit an issues ticket for CQL, please visit the ONC JIRA site
  - <https://oncprojecttracking.healthit.gov/support/projects/CQLIT>

# Questions?

